# F07VUF (CTBCON/ZTBCON) – NAG Fortran Library Routine Document

**Note.** Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

## 1 Purpose

F07VUF (CTBCON/ZTBCON) estimates the condition number of a complex triangular band matrix.

## 2 Specification

```
 SUBROUTINE F07VUF(NORM, UPLO, DIAG, N, KD, AB, LDAB, RCOND, WORK,
1                  RWORK, INFO)
 ENTRY      ctbcon(NORM, UPLO, DIAG, N, KD, AB, LDAB, RCOND, WORK,
1                  RWORK, INFO)
 INTEGER           N, KD, LDAB, INFO
 real              RCOND, RWORK(*)
 complex           AB(LDAB,*), WORK(*)
 CHARACTER*1       NORM, UPLO, DIAG
```

The ENTRY statement enables the routine to be called by its LAPACK name.

## 3 Description

This routine estimates the condition number of a complex triangular band matrix $A$, in either the 1-norm or the infinity-norm:

$$\kappa_1(A) = \|A\|_1\|A^{-1}\|_1 \text{ or } \kappa_\infty(A) = \|A\|_\infty\|A^{-1}\|_\infty.$$

Note that $\kappa_\infty(A) = \kappa_1(A^T)$.

Because the condition number is infinite if $A$ is singular, the routine actually returns an estimate of the **reciprocal** of the condition number.

The routine computes $\|A\|_1$ or $\|A\|_\infty$ exactly, and uses Higham's implementation of Hager's method [1] to estimate $\|A^{-1}\|_1$ or $\|A^{-1}\|_\infty$.

## 4 References

**[1]** Higham N J (1988) FORTRAN codes for estimating the one-norm of a real or complex matrix, with applications to condition estimation *ACM Trans. Math. Software* **14** 381–396

## 5 Parameters

**1:** NORM — CHARACTER*1 *Input*

*On entry:* indicates whether $\kappa_1(A)$ or $\kappa_\infty(A)$ is estimated as follows:

if NORM = '1' or 'O', then $\kappa_1(A)$ is estimated;
if NORM = 'I', then $\kappa_\infty(A)$ is estimated.

*Constraint:* NORM = '1', 'O' or 'I'.

**2:** UPLO — CHARACTER*1 *Input*

*On entry:* indicates whether $A$ is upper or lower triangular as follows:

if UPLO = 'U', then $A$ is upper triangular;
if UPLO = 'L', then $A$ is lower triangular.

*Constraint:* UPLO = 'U' or 'L'.

**3:** DIAG — CHARACTER*1 *Input*

*On entry:* indicates whether $A$ is a non-unit or unit triangular matrix as follows:

if DIAG = 'N', then $A$ is a non-unit triangular matrix;
if DIAG = 'U', then $A$ is a unit triangular matrix; the diagonal elements are not referenced and are assumed to be 1.

*Constraint:* DIAG = 'N' or 'U'.

**4:** N — INTEGER *Input*

*On entry:* $n$, the order of the matrix $A$.

*Constraint:* N $\geq 0$.

**5:** KD — INTEGER *Input*

*On entry:* $k$, the number of super-diagonals of the matrix $A$ if UPLO = 'U' or the number of sub-diagonals if UPLO = 'L'.

*Constraint:* KD $\geq 0$.

**6:** AB(LDAB,$*$) — ***complex*** array *Input*

**Note:** the second dimension of the array AB must be at least max(1,N).

*On entry:* the $n$ by $n$ triangular band matrix $A$, stored in rows 1 to $(k+1)$. More precisely, if UPLO = 'U', the elements of the upper triangle of $A$ within the band must be stored with element $a_{ij}$ in AB$(k+1+i-j,j)$ for $\max(1,j-k) \leq i \leq j$;if UPLO = 'L', the elements of the lower triangle of $A$ within the band must be stored with element $a_{ij}$ in AB$(1+i-j,j)$ for $j \leq i \leq \min(n,j+k)$. If DIAG = 'U', the diagonal elements of $A$ are not referenced and are assumed to be 1.

**7:** LDAB — INTEGER *Input*

*On entry:* the first dimension of the array AB as declared in the (sub)program from which F07VUF (CTBCON/ZTBCON) is called.

*Constraint:* LDAB $\geq$ KD + 1.

**8:** RCOND — ***real*** *Output*

*On exit:* an estimate of the reciprocal of the condition number of $A$. RCOND is set to zero if exact singularity is detected or the estimate underflows. If RCOND is less than ***machine precision***, then $A$ is singular to working precision.

**9:** WORK($*$) — ***complex*** array *Workspace*

**Note:** the dimension of the array WORK must be at least max(1,2$*$N).

**10:** RWORK($*$) — ***real*** array *Workspace*

**Note:** the dimension of the array RWORK must be at least max(1,N).

**11:** INFO — INTEGER *Output*

*On exit:* INFO = 0 unless the routine detects an error (see Section 6).

# 6 Error Indicators and Warnings

INFO $< 0$

If INFO $= -i$, the $i$th parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

# 7 Accuracy

The computed estimate RCOND is never less than the true value $\rho$, and in practice is nearly always less than $10\rho$, although examples can be constructed where RCOND is much larger.

# 8 Further Comments

A call to this routine involves solving a number of systems of linear equations of the form $Ax = b$ or $A^H x = b$; the number is usually 5 and never more than 11. Each solution involves approximately $8nk$ real floating-point operations (assuming $n \gg k$) but takes considerably longer than a call to F07VSF (CTBTRS/ZTBTRS) with one right-hand side, because extra care is taken to avoid overflow when $A$ is approximately singular.

The real analogue of this routine is F07VGF (STBCON/DTBCON).

# 9 Example

To estimate the condition number in the 1-norm of the matrix $A$, where

$$A = \begin{pmatrix} -1.94 + 4.43i & 0.00 + 0.00i & 0.00 + 0.00i & 0.00 + 0.00i \\ -3.39 + 3.44i & 4.12 - 4.27i & 0.00 + 0.00i & 0.00 + 0.00i \\ 1.62 + 3.68i & -1.84 + 5.53i & 0.43 - 2.66i & 0.00 + 0.00i \\ 0.00 + 0.00i & -2.77 - 1.93i & 1.74 - 0.04i & 0.44 + 0.10i \end{pmatrix}.$$

Here $A$ is treated as a lower triangular band matrix with 2 sub-diagonals. The true condition number in the 1-norm is 71.51.

## 9.1 Program Text

**Note.** The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     F07VUF Example Program Text
*     Mark 15 Release. NAG Copyright 1991.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX, KDMAX, LDAB
      PARAMETER        (NMAX=8,KDMAX=NMAX,LDAB=KDMAX+1)
      CHARACTER        NORM, DIAG
      PARAMETER        (NORM='1',DIAG='N')
*     .. Local Scalars ..
      real             RCOND
      INTEGER          I, INFO, J, KD, N
      CHARACTER        UPLO
*     .. Local Arrays ..
      complex          AB(LDAB,NMAX), WORK(2*NMAX)
      real             RWORK(NMAX)
*     .. External Functions ..
      real             X02AJF
      EXTERNAL         X02AJF
*     .. External Subroutines ..
      EXTERNAL         ctbcon
*     .. Intrinsic Functions ..
      INTRINSIC        MAX, MIN
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F07VUF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N, KD
      IF (N.LE.NMAX .AND. KD.LE.KDMAX) THEN
*
```

```
*        Read A from data file
*
         READ (NIN,*) UPLO
         IF (UPLO.EQ.'U') THEN
            DO 20 I = 1, N
               READ (NIN,*) (AB(KD+1+I-J,J),J=I,MIN(N,I+KD))
   20       CONTINUE
         ELSE IF (UPLO.EQ.'L') THEN
            DO 40 I = 1, N
               READ (NIN,*) (AB(1+I-J,J),J=MAX(1,I-KD),I)
   40       CONTINUE
         END IF
*
*        Estimate condition number
*
         CALL ctbcon(NORM,UPLO,DIAG,N,KD,AB,LDAB,RCOND,WORK,RWORK,INFO)
*
         WRITE (NOUT,*)
         IF (RCOND.GE.X02AJF()) THEN
            WRITE (NOUT,99999) 'Estimate of condition number =',
     +        1.0e0/RCOND
         ELSE
            WRITE (NOUT,*) 'A is singular to working precision'
         END IF
      END IF
      STOP
*
99999 FORMAT (1X,A,1P,e10.2)
      END
```

## 9.2 Program Data

```
F07VUF Example Program Data
  4  2                                                    :Values of N and KD
  'L'                                                     :Value of UPLO
 (-1.94, 4.43)
 (-3.39, 3.44) ( 4.12,-4.27)
 ( 1.62, 3.68) (-1.84, 5.53) ( 0.43,-2.66)
               (-2.77,-1.93) ( 1.74,-0.04) ( 0.44, 0.10)  :End of matrix A
```

## 9.3 Program Results

```
 F07VUF Example Program Results

 Estimate of condition number =  3.35E+01
```